# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=factordb-pycli
_pkgname=factordb-python
pkgver=1.3.0
pkgrel=8
pkgdesc='CLI for factordb and Python API Client.'
arch=('any')
groups=('blackarch' 'blackarch-crypto')
url='https://github.com/ryosan-470/factordb-pycli'
license=('MIT')
depends=('python' 'python-requests')
makedepends=('python-setuptools' 'python-pip' 'python-wheel')
source=("https://github.com/ryosan-470/factordb-python/archive/$pkgver.tar.gz")
sha512sums=('832b0baf749fc43befb68b1b312ad498aec839b45bf98a9ebf6131d13595e7aa0abc228b2737aeaf4f9d5eef4f8e9f188137fb0dd61dcb831c38505461a6d024')

prepare() {
  cd "$_pkgname-$pkgver"

  cp --no-preserve=ownership README.md README.rst
}

build() {
  cd "$_pkgname-$pkgver"

  python setup.py build
}

package() {
  cd "$_pkgname-$pkgver"

  python setup.py install --root="$pkgdir" --prefix=/usr -O1 --skip-build
}

